Utilidad BUSCA001
Esta utilidad permite buscar un miembro determinado entre las distintas librerías que tiene allocadas el usuario.                               

Esta opción tiene en cuenta el orden de allocación de las librerías     

Además esta utilidad combina los 2 lenguajes de comandos REXX y Clist

	------------------------------------------------------------------------------ 

 _    JMP                  Buscar Modulo                   USUARIO: USRJMP    _   

 ------------------------------------------------------------------------------ 

 Command ==>  __________________________________________________________________
        Esta utilidad permite buscar un objeto entre las distintas              

        librerías que tiene allocadas el usuario.                               

        Esta opción tiene en cuenta el orden de allocacion de las librerías     

                  Cadena...... ===> JMP_____                                    

        Fichero con librerías  ===> ENVDES.USRJMP.AUX____________________         

    Pulse  INTRO  para continuar  F3  para salir                                




	**********************************************************                     

 Objeto  JMP                                                                    

 encontrado en                                                                  

 ENVDES.USRJMP.PNL                                                                

 ENVDES.USRJMP.CMD                                                                

 *********************************************************                      

 ***                                                                            




1 REXX: BUSCA001        
	/****************************    REXX     *****************************/

/**********************************************************************/

/*         Buscar objeto entre librerias allocadas por usuario        */

/**********************************************************************/

Address Ispexec "CONTROL ERRORS RETURN"

Address TSO "Profile Noprefix"

debug           = 'SIx'

/* entrada = 'PRUEBAS'  */                       /* Solo para pruebas */

/*0.  Inicializacion de variables  ----------------------------------**/

prog_arq        = 'BUSCA011'

ENT_INVOC       = 'ONLINE'

entorno         = 'DES'

MENSAJE1        = ''

MENSAJE2        = ''

MENSAJE3        = ''

MENSAJE4        = ''

USUARIO         = USERID()

FICHERO         = 'ENVDES.'usuario'.AUX'

buscar:

/*****/

permiso = 'ABIERTO(SI)'

if permiso = 'ABIERTO(SI)' then signal inicio

else

do

   say

   say '********************************************************'

   say '**             OPCION CERRADA TEMPORALMENTE           **'

   say '**    PARA CUALQUIER ACLARACION CONTACTE CON USRJMP   **'

   say '********************************************************'

   exit

end

/**********************************************************************/

/* Comienza la ejecución de la aplicación                             */

/**********************************************************************/

Inicio:

/*****/

/**********************************************************************/

/** Parte 1:                                                ---------**/

/** --------------- Ciclo para la pantalla busca001  ----------------**/

/** Se pide:                                                         **/

/**    Nombre del objeto a buscar                                    **/

/**    Nombre del fichero de entrada (Listado de librerias allocadas)**/

/**********************************************************************/

ADDRESS ISPEXEC "VGET objeto SHARED"

DO FOREVER

   ADDRESS ISPEXEC " DISPLAY PANEL(BUSCA001)"
   rc_panel = rc

   /* Respuesta del panel busca001 correcta  ----------------------*/

   IF rc_panel = 0 THEN DO

      UPPER objeto

      UPPER fichero

      /****************************************************************/

      /** --------- Cargar las librerias alocadas    ----------------**/

      /****************************************************************/

      address tso "ex 'ENVDES.USRJMP.cmd(cargaalo)' '"fichero"'"

      /*01.3 Existe el fichero de entrada?  --------------------------*/

      call existe_fichero

      zcursor = 'objeto'

      IF RESULT = 98 THEN ITERATE

      LEAVE  /* TODO BIEN ---> salgo del ciclo y comienzo proceso  ---*/
   END /* Fin del: IF rc_panel=0  ==> la parte del THEN   busca001  --*/
   /* Respuesta del panel busca001 incorrecta  -----------------------*/

   ELSE DO

      IF rc_panel = 8 THEN do         /* Se dio F3  ------------------*/

         EXIT                         /* Termina la ejecucion   ------*/

      END

      ELSE DO                         /* Otro RC = Error en el panel  */

         TIPMSG   = ' DE ERROR '||'/'

         MENSAJE1 = prog_arq    ||'/'

         MENSAJE2 = ' ERROR al visualizar el PANEL BUSCA001.   '||'/'

         MENSAJE3 = ' El codigo de retorno es ' RC '.          '||'/'

         MENSAJE4 = ' CONTACTE CON USRJMP.'

         CALL ERROR_GRAL

      END

   END   /* Fin del:  IF rc_panel=0  De la pantalla  BUSCA001  ------ */
END      /* Fin del:  DO FOREVER     De la pantalla  BUSCA001  -------*/
/**********************************************************************/

/** Parte 2:                                                ---------**/

/*--------------------------------------------------------------------*/

/*          tratar el fichero de entrada                    ----------*/

/**********************************************************************/

/*----------abrir fichero de entrada ---------------------------------*/

   "ALLOC DA("fichero") F(listado) SHR REUSE"

   "EXECIO * DISKR listado (FINIS STEM linea."

   lista = ""

/* el primer elemento del fichero de entrada es un comentario */

   DO i=2 TO linea.0

      tabla.i     = SUBSTR(linea.i,1,70)

      UPPER tabla.i

      elemento = strip(tabla.i)

      /* -------------------------------------- */

      /* las librerias sys1.xxx no son publicas */

      /* el terminal no se examina              */

      /* -------------------------------------- */

      parse var elemento raiz'.'pro1'.'pro2

      if debug = "SI" then say raiz '-' pro1 '-' pro2

      if  left(elemento,9) ¬= "SYS1.UADS" ,

      & right(elemento,2)  ¬= ".?" ,

      & raiz               ¬= "SYS1" ,

      & elemento           ¬= "TERMFILE" then do

        pds_name = elemento

        call examinar_libreria
      end

      if debug = "SI" then if i = 5 then leave

   END

/* ------------------------------------------------------------------ */

/*                desglosar lista de encontrados                      */

/* ------------------------------------------------------------------ */

   x = lista

   if x ¬= '' then

     do

        say '**********************************************************'

        say 'Objeto ' objeto

        say '                                    '

        say 'encontrado en '

        say '                                    '

        do j = 1 to words(x)

           say word(x,j)

        end

        say '                                    '

        say '*********************************************************'

     end

   else

     do

        say '*********************************************************'

        say 'programa ' objeto ' no encontrado en las librerias  '

        say '                                    '

        say 'contenidas en el fichero ' fichero

        say '                                    '

        say '*********************************************************'

     end

return

/**********************************************************************/

/**  RUTINAS       RUTINAS       RUTINAS       RUTINAS       RUTINAS **/

/**         RUTINAS       RUTINAS       RUTINAS       RUTINAS        **/

/**********************************************************************/

/*********************************************************************/

existe_fichero:

/**********************************************************************/

x = outtrap('On')

ADDRESS TSO "LISTD " fichero

x = outtrap('Off')

IF RC = 12 | RC = 20 THEN DO

   TIPMSG   = ' DE ERROR  '||'/'

   MENSAJE1 = prog_arq     ||'/'

   MENSAJE2 = ' No existe en HOST el fichero:    '||'/'

   MENSAJE3 = fichero     ||'/'

   MENSAJE4 = ' Por favor, introduzca el nombre correcto.'

   CALL ERROR_PARCIAL
   RETURN 98

END

RETURN

/**********************************************************************/

ERROR_PARCIAL:

/**********************************************************************/

/* En principio siempre va a entrar por la 1º parte del IF  ----------*/

/* porque ENT_INVOC es igual a ONLINE    -----------------------------*/

IF ENT_INVOC = 'ONLINE' THEN DO

   CALL RUTGXERR TIPMSG MENSAJE1 MENSAJE2 MENSAJE3 MENSAJE4

END

ELSE DO

   CALL RUTGXERB TIPMSG MENSAJE1 MENSAJE2 MENSAJE3 MENSAJE4

   ZISPFRC = 23

   ADDRESS ISPEXEC "VPUT ZISPFRC SHARED"

END

/* Diferencia con la rutina de ERROR_GRAL:           -----------------*/

/* RETURN y no EXIT 23 para que continue en la REXX  -----------------*/

RETURN

/**********************************************************************/

ERROR_GRAL:

/**********************************************************************/

IF ENT_INVOC = 'ONLINE' THEN DO

   CALL RUTGXERR TIPMSG MENSAJE1 MENSAJE2 MENSAJE3 MENSAJE4

END

ELSE DO

   CALL RUTGXERB TIPMSG MENSAJE1 MENSAJE2 MENSAJE3 MENSAJE4

   ZISPFRC = 23

   ADDRESS ISPEXEC "VPUT ZISPFRC SHARED"

END

CALL liberar_librerias
EXIT 23

/*--------------------------------------------------------------------*/

/* el siguiente parrafo memoriza el directorio de cada libreria       */

/* lista los miembros de un pds                                       */

/* Las 6 primeras lineas contienen informacion del PDS                */

/* a partir de la septima linea se encuentran listados los miembros   */

/**********************************************************************/

examinar_libreria:

   LibPgms   = pds_name

   obj_name  = objeto

   y = outtrap('miembro.')

   address tso "listd ("LibPgms") members"

   y = outtrap('off')

   if rc ¬= 0 then return

   if debug = "SI" then say ' rc:' rc ':' pds_name

   /*----------------------------------------------------*/

   /* examinar si existe un miembro con ese nombre ------*/

   /* en la tabla de miembros del pds de fuentes   ------*/

   /*----------------------------------------------------*/

   m = 6

   do while m < miembro.0

       m = m + 1

       miembro.m = strip(miembro.m)

       if strip(obj_name) =  miembro.m then do

          /*----------examinar si ya esta en la lista de encontrados */

          existe = "NO" then do

          do z = 1 to words(lista)

             if Strip(word(lista,z)) = strip(pds_name) then do

                existe = "SI"

                leave

             end

          end

          /*----------incluirle en la lista de encontrados ----------*/

          if existe ¬= "SI" then do

             lista = lista || ' ' || pds_name

             if debug = "SI" then do

                say   '*****************************'

                say   '@El objeto ' obj_name ' existe en ' pds_name

             end

             leave

          end

       end

   end

   /*----------------------------------------------------*/

return







2 Panel: BUSCA001

	)ATTR

  _ TYPE(INPUT) INTENS(HIGH) PAD('_')

  ¬ TYPE(INPUT) INTENS(HIGH) PAD('_')  CAPS(ON)

  | TYPE(&TIPOS) INTENS(&VISIBLE)

  $ TYPE(OUTPUT) INTENS(HIGH)

  { TYPE(TEXT) INTENS(LOW) COLOR(TURQ) HILITE(REVERSE) SKIP(ON)

  ] TYPE(TEXT) INTENS(HIGH) COLOR(GREEN) SKIP(ON)

  [ TYPE(TEXT) INTENS(HIGH) COLOR(BLUE) SKIP(ON)

  ? TYPE(&TIPO) INTENS(HIGH)

  } TYPE(OUTPUT) INTENS(NON ) COLOR(RED  )

)BODY WIDTH(80)

%

%------------------------------------------------------------------------------

% {   JMP                  Buscar Modulo                   USUARIO: &ZUSER  %

%------------------------------------------------------------------------------

%

%Command ==> _ZCMD

%

%

%      [Esta utilidad permite buscar un objeto entre las distintas

%      [librerias que tiene allocadas el usuario.

%      [Esta opcion tiene en cuenta el orden de allocacion de las librerias

%

%

]                 Cadena......%===>¬Z       ]

%

]       Fichero con librerias %===>¬Z                                  ]

%

%

%

%

%   Pulse ]INTRO %para continuar ]F3 %para salir

)INIT

  &ZCMD = ' '

  &TIPOS = 'OUTPUT'

  &VISIBLE = 'NON'

  &SUBSYS  = 'DS0D'

  .HELP = ' '

  .ZVARS = '(OBJETO,FICHERO)'

  .CURSOR = OBJETO

  IF (&ZUSER = 'USRJMP1','USRJMP')

     &TIPOS = 'INPUT'

     &VISIBLE = 'HIGH'

)PROC

  VER (&OBJETO,NB,MSG=JMPP002)

  VER (&FICHERO,NB,MSG=JMPP002)

  VPUT (OBJETO) SHARED

  VPUT (FICHERO) SHARED

)END




3 Clist: CARGAALO

	 /* <<<<<<<<<<<<<<<<<<<<<<  CARGAALO >>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>  */

 /*                                                                   */

 /*             CAPTURA LAS LIBRERIAS ALOCADAS                        */

 /*             Y LAS ESCRIBE EN EL FICHERO                           */

 /*             QUE SE RECOGE POR PARAMETRO                           */

 /*                                                                   */

 /* ================================================================= */

         PROC 1 FICHERO

         CONTROL MSG

         SET CABECERA = <<< LIBRERIAS ALOCADAS >>>

         SET RC = 0

 /*      EXEC 'XE03029.DE4JPP.CMD(JMPSEG01)','A'                      */

         CONTROL NOLIST NOCONLIST

         SET &SYSOUTTRAP = 999

         LISTA

         SET &SYSOUTTRAP = 0

         SET &NOMBRE = TABLA

         SET &CONTA  = 0

         SET &ULTIMO = &SYSOUTLINE

 /*                                                                   */

 /*      ============= MEMORIZAR SALIDA ============================= */

 /*                                                                   */

 LOOP1:  SET &CONTA  = &CONTA + 1

         IF  &CONTA > &ULTIMO THEN GOTO LISTAR

         SET    &NOMBRE&CONTA = &&SYSOUTLINE&CONTA

         GOTO LOOP1

 EXIT1:  SET &CONTA  = 0

         WRITE NUMERO DE ELEMENTOS: &ULTIMO

 /*                                                                   */

 LOOP2:  SET &CONTA  = &CONTA + 1

         IF  &CONTA > &ULTIMO THEN GOTO LISTAR

         SET    &SALIDA = &&NOMBRE&CONTA

         WRITE  &SALIDA

         GOTO LOOP2

 /*                                                                   */

 /*      SET &FICHERO= &SYSUID..DE4JPP.AUX */

 LISTAR: SET &FF = 'FF'

 /* TAR: SET &FICHERO= ENVDES.&SYSUID..AUX */

 /*      WRITE                                                        */

 /*      WRITENR LA SALIDA CONTIENE &ULTIMO LINEAS                    */

 /*      WRITE                                                        */

 /*                                                                   */

 /*      ============= ESCRIBIR SALIDA EN FICHERO =================== */

 /*                                                                   */

         IF &SYSDSN('&FICHERO') = OK THEN +

              ALLOC FILE(SALIDA) DATASET('&FICHERO') OLD REUSE

         ELSE ALLOC FILE(SALIDA) DATASET('&FICHERO') NEW CATALOG +

                                 UNIT(3380) RECFM(F B) LRECL(80) +

                                 BLKSIZE(6240) TRACKS SPACE(3 1) REUSE

 /*                                                                   */

         OPENFILE SALIDA OUTPUT

         SET     &SALIDA = 'LA SALIDA CONTIENE:'

         PUTFILE  SALIDA

 /*                                                                   */

         SET &RC = 0

         ERROR +

             DO

                SET &RC = &LASTCC

                IF &RC = 400 THEN +

                   WRITE FIN NORMAL DE LA CLIST

                ELSE  +

                   DO

                      WRITE INEXPERADO FIN DE CLIST

                      WRITE RC ES &RC

                      WRITE LINEA

                      EXIT

                   END

              END

 /*                                                                   */

         SET     &CONTA  = 6

 GRABAR: SET     &CONTA  = &CONTA + 1

         IF      &CONTA  > &ULTIMO THEN GOTO CERRAR

         SET     &LINEA  = &&NOMBRE&CONTA

         SET     &SALIDA = &LINEA

         PUTFILE  SALIDA

 /*      WRITE   &SALIDA                                              */

         GOTO    GRABAR

 CERRAR: CLOSFILE SALIDA

         FREE FILE(SALIDA)

